
EXCEL - FORMULES DE CALCUL : AIDE-MÉMOIRE  
DES OPÉRATEURS,  DES POINTEURS  ET  DES FONCTIONS 

I -  OPÉRATEURS 

Opération Opérateur Exemples et remarques 

Addition + 
=B4+B$5              =TotalHT+FraisPort 

On peut utiliser l’adressage de type « bataille navale » (ligne/colonne)  
(avec ou sans dollars $),  

mais il est souvent plus lisible d’utiliser les noms de cellules 

Soustraction - =B4-B$5              =TotalHT-Remise 

Multiplication  * =B4*B$5              =TotalHT*TauxTVA*1,01 

Division / =B4/B$5              =Total/B3 

Les 4 opérateurs arithmétiques sont en double sur le clavier : pavé alpha et pavé numérique. 

Puissance ^ 
Le signe ̂  est saisi en tapant AltGr 9  (sur le pavé alpha) 
Exemple :          10²     s’écrit     10^2      sous Excel. 

Changement 
de signe - En plaçant un « moins » devant une expression, on en change le signe. 

Domaines de 
cellules 

: 

; 

On peut désigner des groupes de cellules : 
Le signe deux-points ( : ) dit « jusqu’à ».  

Par exemple,  A1:C4  dit  « toutes les cellules de A1 jusqu’à C4 ». 
Le signe point-virgule ( ; )  signifie « et ». 

Par exemple,  A1;C4  dit  « les cellules A1 et C4 ». 

Formules 3D : 

Quand on travaille sur une seule feuille, on parle de 2D (lignes/colonnes). 
Quand on ajoute des liens vers d’autres feuilles, on utilise une 3ème dimension.  
Les formules qui joue de cette possibilité sont des formules "3D". Par exemple, 
on peut additionner toutes les cellules C3 des feuilles Janvier à Décembre dans 
la cellule C3 d’une 13ème feuille : on écrira : 

=SOMME('Janvier : Décembre' ! C3) 

Ordre  
de calcul  

Sans parenthèses, Excel effectue les opérations dans l’ordre suivant : 
 – Négation (comme dans –1) 
 % Pourcentage 
 ^ Exposant 
 *  et  / Multiplication et division  
 +  et  –  Addition et soustraction 
 & Concaténation de deux chaînes de texte  
=  <  >  <=  >=  <> Comparaison 

Ordre forcé (  ) 
Avec les parenthèses, on impose l’ordre des opérations. 
L’opération =3*2+10 donne 16   (XL fait d’abord la multiplication),   
alors que =3*(2+10) donne 36… (on force XL à commencer par l’addition) 
  Ceci est essentiel. 

Comparaisons 

> 

>= 

< 

<= 

= 

<> 

Les fonctions SI principalement, mais aussi quelques commandes et autres ou-
tils, utilisent un des opérateurs de comparaison pour exprimer une condition : 
ci-contre, on lit :  
 supérieur     > 
 supérieur ou égal     >=  
 inférieur    < 
 inférieur ou égal     <= 
 égal    = 
 différent   <>. 

Opération Opérateur Exemples et remarques 

Figer une 
adresse pour 
la prochaine 

recopie 

$ 

Lors d’une recopie de formule,  
Excel transforme tout seul la formule  

pour qu’elle fonctionne encore à sa nouvelle place… 

Par exemple, si on recopie vers le bas,  
Excel change la partie « ligne » des adresses de cellules dans la formule,  

pour que la formule fonctionne en s’adaptant à sa nouvelle ligne. 

Quand une partie de la formule pointe vers une cellule de référence qui doit 
être la même dans toutes les formules (TauxTVA par exemple), il faut bloquer 
le mécanisme de transformation automatique de l’adresse qu’excel exécute 

lors de la recopie : en plaçant le signe dollar $ devant un numéro de colonne ou 
un numéro de ligne, on « fige » ce numéro : 

Par exemple, $A$1 est une adresse complètement figée :  
on peut la recopier dans tous les sens, elle restera $A$1… 

Autre exemple :  
Dans A$1, seule la ligne (1) est figée, pas la colonne (pas le A) :  

si on recopie vers le bas, le 1 reste 1,  mais si on recopie vers la droite,  
le A devient B, C, D, etc.       La touche utile est  F4 ! 

C’est précieux pour ne créer qu’une seule formule pour tout un tableau : 
par exemple, pour calculer des pourcentages ci-dessous, on ne crée qu’une 

seule formule, utilisant astucieusement le dollar $. Puis on la recopie d’abord la 
cellule vers le bas, puis toute la colonne vers la droite : 

 

Concaténation & 

Précieux opérateur qui permet d’« additionner des textes » :  
Si je veux qu’une cellule agrège, compile, rassemble le prénom et le nom 
contenus dans des cellules séparées, je crée la formule : =Prénom&Nom 

 
Pour ajouter des constantes (par exemple, des espaces),  
on les encadre de guillemets " " . 
 

Exemple : =B3&"  "&A3  permet de ne pas coller le nom et le prénom. 

II -  POINTEURS 

A -  CROIX BLANCHE POUR SÉLECTIONNER 

La grosse croix blanche  apparaît quand on vise le corps des cellules.  

Elle signale qu’Excel est prêt à  sélectionner des cellules  (les mettre en surbrillance)  dès 
qu’on va cliquer ou glisser avec la souris :   

 

Pour bien comprendre à quel moment le pointeur change d’apparence, il faut intellectuellement ré-
duire la croix à son centre.  Imaginez que la croix blanche a un centre rouge lumineux…   Il n’y a 
que cette lumière rouge qui compte… 

B -  FLÈCHE BLANCHE POUR DÉPLACER OU COPIER 

La flèche blanche    apparaît quand on vise le bord noir d’une sélection.  

Elle signale qu’Excel est prêt à  déplacer/copier des cellules dès qu’on va glisser avec la 
souris  (avec ou sans la touche Ctrl) : 

 

Intellectuellement, il faut réduire la flèche à sa pointe. Imaginez que la pointe a un bout rouge lu-
mineux… Il n’y a que cette pointe lumineuse qui compte… Vous comprendrez mieux à quel mo-
ment le pointeur change d’apparence. 

 
Notez bien : sélectionner n’est pas agir !  

 
Si vous avez mal sélectionné,  

il ne faut pas essayer  
d’annuler une sélection avec Ctrl Z   

 
(vous annuleriez la commande précédente,  

ce qui n’a rien a voir…). 
 

Non, en cas d’erreur de sélection,  
il suffit de sélectionner à nouveau :  

clic ou glissade avec la grosse croix blanche,  
 

et « une sélection chasse l’autre »…  

C -  CROIX NOIRE POUR COPIER EN INCRÉMENTANT 

La croix noire    apparaît quand on vise le coin bas droit d’une sélection.  

Elle signale qu’Excel est prêt à  copier la sélection en incrémentant dès qu’on va glis-
ser avec la souris : 

 

Intellectuellement, il faut réduire la croix à son centre.    Imaginez que la croix noire a un centre 
rouge lumineux… Il n’y a que cette lumière rouge qui compte… Vous comprendrez mieux à quel 
moment le pointeur change d’apparence. 

D -  i NOIR POUR PLACER LE POINT D’INSERTION 

Le i majuscule noir  
  
apparaît quand on vise une zone de saisie de texte   (la barre de for-

mule, souvent).   Il signale qu’Excel est prêt à  placer le point d’insertion  là où on va cli-
quer avec la souris : 

  

Intellectuellement, il faut réduire le i noir à son centre.  Imaginez que le i a un centre rouge lumi-
neux… Il n’y a que ce point lumineux qui compte… Vous comprendrez mieux à quel moment le 
pointeur change d’apparence. 

Attention :  
Cet outil de recopie écrase tout sur son passage : 
je l’appelle recopie « rouleau compresseur » … 

 
Par ailleurs,  

Excel n’incrémente que quand il le peut  
(sinon il recopie mais n’incrémente pas) :  

il connaît lundi, mardi… janvier, février… 
il est capable de prolonger la suite Région 1… 

il est même capable d’utiliser vos suites personnelles  
(créées avec la commande Outils Options,  

onglet Listes pers.)… 
 

Mais quelquefois,  
il faut sélectionner 2 cellules de départ   

pour qu’Excel puisse deviner la suite :  

E -  DOUBLE TRAIT - DOUBLE FLÈCHE  
POUR LES RÉGLAGES DE LARGEURS /HAUTEURS 

Le pointeur double trait-double flèche       apparaît quand on vise une zone de réglage de 
volet   (entre deux zones « à géométrie variable »…).   Il signale qu’Excel est prêt à  déplacer le 

trait de séparation  à partir de là où on va glisser avec la souris : 

 

Comme d’habitude, pour comprendre les transformations de votre pointeur, réduisez-le in-
tellectuellement à son centre de gravité  (son centre, souvent…)… 

 

F -  POINTEURS SUR OBJETS GRAPHIQUES 

Enfin, sur tous les objets dessinés (les formes automatiques, les graphiques, les images impor-
tées, les photos, les organigrammes, les zones de texte, les WordArts, etc.),  
le pointeur peut prendre deux apparences, suivant que l’on vise le corps de l’objet ou ses poignées 
(8 carrés placés autour de lui après avoir cliqué dessus)  :  

Soit c’est le pointeur « 4 points cardinaux »  

qui permet de déplacer/copier l’objet graphique 
(pour copier, il faut glisser en gardant Ctrl  enfoncée) : 

 

Soit c’est le pointeur double flèche noire  

qui permet de redimensionner l’objet : 

 

 

Largeur et hauteur des lignes et des colonnes 

Tailles respectives de  
la zone des onglets et de  
celle des ascenseurs 

Création, réglage et suppression  
d’un volet 

III - FONCTIONS CHOISIES  (BEST OF…) 

Les fonctions d’Excel sont des mots réservés que l’on peut taper dans une formule pour obtenir fa-
cilement un résultat élaboré.  Je vous présente ici  les 50 fonctions les plus importantes. 

Toutes les fonctions d’Excel utilisent des parenthèses.  

Entre ces parenthèses, on précise les contraintes du calcul : Excel appelle arguments ces informa-
tions entre parenthèses. 

Les arguments sont séparés par le signe point-virgule   ;  

Certaines fonctions n’ont pas besoin d’arguments : on tape alors 2 parenthèses collées.  
D’autres n’ont besoin que d’un seul argument : on n’a pas besoin de point-virgule dans ce cas. 

J’ai souvent mis des espaces autour des points-virgules et des parenthèses pour une bonne lisibilité , 
mais attention : il ne faut JAMAIS taper d’espace dans les formules !  
 

Objectif Fonction, syntaxe et exemples 

Pour additionner de nombreuses cellules,  
sans avoir à les citer toutes dans la formule,  
il faut utiliser la fonction SOMME,  
et décrire la plage de cellules à sommer.  

Le signe deux-points :    dit  « jusqu’à »,  
le signe point-virgule  ;   dit  « et ». 

Syntaxe :  
SOMME ( plage ) 

Exemples :  
SOMME ( B2 : B30 ) 

SOMME ( MontantHT ) 
SOMME ( TotalSem1 ; TotalSem2 ) 

 

Avantages de la fonction SOMME  
sur l’opérateur + : 

• Ignorer  les cellules contenant du texte. 
Avec +, on a un message d’erreur si une des cel-
lules sommées contient un texte !. 

• Décrire rapidement un grand nombre de 
cellules.  
Avec +, la saisie est longue et fastidieuse. 

• Décrire la plage de façon fiable, sans oubli.  
Avec l’opérateur +, le risque d’erreur est grand : 
oubli, doublon, erreur de pointage. 

• Permettre des insertions futures au cœur 
de la plage sommée.  
Avec +, les futures insertions ne seront pas 
concernées par l’addition. 

Astuce  -  Conseil 

Vous allez être tenté de placer vos sommes en dessous des 
données de base, même si vos données de base sont régulière-
ment augmentées de saisies nouvelles.   
Ceci va vous obliger à insérer des lignes chaque fois que vous 
aurez des saisies à opérer. 

Il est beaucoup plus malin de placer les sommes en tête, au-
dessus des données de base, et en visant une plage beaucoup 
plus vaste que la zone des données actuellement saisies (voir 
exemple ci-dessus) : toute la colonne par exemple (sans 
prendre la formule elle-même, attention !!!). 

Vous aimerez sûrement voir immédiatement vos données de 
synthèse, en haut des colonnes, dès que vous entrez dans le 
classeur, et dans le volet haut, même pendant vos déplace-
ments et corrections des données de base !    Essayez ! vous 
verrez… 

Objectif Fonction, syntaxe et exemples 

Pour classer vos données, en plus de la com-
mande Données Trier qui agit ponctuellement, 
sur demande, on peut déterminer le rang d’une 
cellule dans une plage de cellules avec la fonc-
tion  RANG.  

On calculera le rang de la première cellule, puis 
on recopiera cette formule pour calculer le rang 
des autres.  

Mais il faudra faire attention à figer la plage de 
référence (avec des dollars, touche F4) pour 
qu’elle reste la même dans toutes les copies.  

 

 

 

 

 

 

Vous observez sur cet exemple que la fonction 
RANG est souvent utilement complétée par 
la commande Format Mise en forme condi-
tionnelle qui permet, ici, d’afficher automati-
quement les 10 premiers en fond bleu, et les 
derniers (après le rang 20) en rouge… Les au-
tres restant sur fond blanc… 

Syntaxe :   RANG ( cellule ; plage ) 
Exemple :  

 
 

 

 

 

 

 

Dans le même ordre d’idées (classement), on 
évoquera des fonctions simples qui permettent, 
à partir d’une plage de cellules, d’extraire la 
plus grande, la plus petite, le nombre de valeurs, 
la moyenne… 

Fonctions  
MIN, MAX, NB, NBVAL, MOYENNE 

En jouant d’astuce, on peut décupler les per-
formances d’une fonction : par exemple,  
pour plafonner un calcul, je vais utiliser  
MIN ( calcul ; plafond ).   
De même, pour ne pas dépasser un plancher, 
je vais utiliser MAX ( calcul ; plancher ). 

Exemples : 
MIN ( base * taux ; PlafondSécu )   
pour plafonner un calcul de cotisation sociale. 
MAX ( CA * 0,10 ; Fixe ) pour ne jamais 
payer un vendeur moins que le minimum fixe 
prévu   (plancher).  

Syntaxe : MIN ( plage ) 
MAX ( plage ) 
MOYENNE ( plage )  
NB ( plage ) 
NBVAL ( plage ) 

Exemples :  

MAX ( B2 : B120 ) sort le plus grand nombre. 

MOYENNE ( Notes ) calcule la moyenne. 

NB ( C5 : C37 ) calcule le nombre de valeurs  
numériques  en ignorant les textes  éventuels. 

NBVAL ( C5 : C37 ) calcule le nombre de cellules  
non vides,  en comptant les textes et les chiffres. 

Objectif Fonction, syntaxe et exemples 

Pour saisir et manipuler des dates et des délais, il 

faut connaître les  fonctions date : 
Excel utilise un calendrier où chaque jour est un 
rang par rapport au 1er janvier 1900.  

C’est ainsi que le 9 juillet 2001 est le chiffre 
37 081 mis au format  j mmmm aaaa… 

Par ailleurs, les heures sont naturellement des 
fractions de jour : 12h (midi) est 0,5, 18h est 0,75, 
6h du matin est 0,25, etc. 

Donc, le lundi 9 juillet 2001 à 6h se dit 37 081,25 
au format jjjj j mmmm aaaa " à " h"h" 

On utilise les séparateurs / ou – (slash ou tiret) 
pour saisir les dates.  On utilise le séparateur : 
(deux points) pour saisir les heures. 

Exemple : il suffit de saisir 9/7 pour le lundi 9 juillet 
2001, car Excel devine que sans année spéci-
fiée, c’est l’année en cours qu’il faut prendre. 

Autre exemple : pour saisir 6h, il faut taper 6:0 
(c’est le :0 qui signale à Excel qu’il s’agit d’une heure). 

Soustraire deux dates donne un délai. 

Additionner ou soustraire une date et un délai 
donne une date 

Additionner deux dates est absurde. Multiplier 
ou diviser 2 dates également. 

Pour découvrir seul l’intégralité des fonctions offer-
tes par Excel, appelez l’Aide en ligne , pas avec 
la touche F1 : le trombonne est totalement incom-
pétent, il ne comprend rien aux questions : utilisez 
plutôt la commande ? (point d’interrogation) 
Sommaire et Index, à droite du menu, et cherchez 
le mot fonctions, de date… 

Même éteint, l’ordinateur mémorise la date et l’heure (avec une pile)  

On peut donc récupérer  « la date-sytème »  
et  « l’heure-système » : 

=AUJOURDHUI ( ) rend la date sytème (pas l’heure),  
donc un nombre entier. 

=MAINTENANT ( ) rend la date et l’heure système,  
donc un nombre décimal. 

=DATE ( année ; mois ; jour ) rend le nombre correspondant 
à la date décrite.  Par exemple, DATE(01;07;09) rend 37 081.  
Avec cette fonction, on peut créer une date de toute pièces, 
morceau par morceau…  

Par exemple : =DATE ( AnnéeAchat + 1 ; MoisAchat ; 1 )  
permet de calculer la première date anniversai-
re d’un achat, au 1er du mois. 

=JOUR ( date ) ,    MOIS ( date )    et    ANNEE ( date )  
permettent d’extraire une partie de la date,  
pour effectuer des tests ou recomposer une autre date. 

Les formats personnalisés à connaître sont : 

 Code Format Exemple 
 j jour sur 1 chiffre 9 
 jj jour sur 2 chiffres 09 
 jjj jour sur 3 lettres lun 
 jjjj jour en toutes lettres lundi 
  

 m mois sur 1 chiffre 7 
 mm mois sur 2 chiffres 07 
 mmm mois sur 3 ou 4 lettres juil 
 mmmm mois en toutes lettres juillet 
  

 aa année sur 2 chiffres 01 
 aaaa année sur 4 chiffres 2001 
 

 h heure sur 1 chiffre 6 
 hh heure sur 2 chiffres 06 
 m minutes sur un chiffre 0 
 mm minutes sur 2 chiffres 00 

 Vous noterez que ET ( cond1 ; cond2 ; cond3…)  et  OU ( cond1 ; cond2 ; cond3…) sont des fonctions à utiliser à l’intérieur du pre-
mier argument des fonctions SI.      Un bloc "ET" ou un bloc "OU" génère VRAI ou FAUX.       Le SI s’en sert ensuite pour agir… 

Objectif Fonction, syntaxe et exemples 

La fonction DATEDIF    (pour les calculs d'âge)   (voir le site de Laurent Longre sur  http://longre.free.fr/) 

DATEDIF fait partie des fonctions masquées d'Excel. Elle n'est pas référencée par l'aide en ligne. Elle est pourtant 
bien commode… 

DATEDIF permet de  calculer la différence entre deux dates en années, mois et jours.  

Syntaxe : =DATEDIF ( Date1 ; Date2 ; Intervalle )  
Cette fonction renvoie la différence entre Date1 et Date2 (Date2 >= Date1) selon l'argument Intervalle, qui peut 
prendre les valeurs suivantes :  
 "y" : différence en années 
 "m" : différence en mois 
 "d" : différence en jours 
 "ym" : différence en mois, une fois les années soustraites 
 "yd" : différence en jours, une fois les années soustraites 
 "md" : différence en jours, une fois les années et les mois soustraits  

Exemple : =DATEDIF ( "5/4/1990" ; "15/8/99" ; Intervalle )  
renvoie les valeurs suivantes selon la valeur de l'argument Intervalle :  
"y" : 9 (ans) 
 "m" : 112 (mois) 
"d" : 3419 (jours) 
"ym" : 4 (mois restants, une fois les 9 ans soustraits) 
"yd" : 132 (jours restants, une fois les 9 ans soustraits) 
"md" : 10 (jours restants, une fois les 112 mois soustraits)  

La fonction DATEDIF peut être en particulier utilisée pour calculer des âges.  
Par exemple, si la cellule A1 contient une date de naissance et la cellule B1 la date du jour :  

- Âge en années simples :  
=DATEDIF(A1;B1;"y") & SI(DATEDIF(A1;B1;"y")>1;" ans";" an")  

- Âge en années et mois :  
=DATEDIF(A1;B1;"y") & SI(DATEDIF(A1;B1;"y")>1;" ans, ";" an, ") & DATEDIF(A1;B1;"ym") & " mois"  

- Âge en années, mois et jours :  
=DATEDIF(A1;B1;"y") & SI(DATEDIF(A1;B1;"y")>1;" ans, ";" an, ") & DATEDIF(A1;B1;"ym") & " mois, " 
& DATEDIF(A1;B1;"md") & SI(DATEDIF(A1;B1;"md")>1;" jours";" jour")  

On peut utiliser quelques autres fonctions de date très pratiques, mais pas disponibles tout de suite…  
Il faut d’abord installer des modules additionnels :  
lancer Outils Macros complémentaires, et ajouter Utilitaire d’analyse, s’il n’est pas déjà en place… 

MOIS.DECALER ( date_départ ; mois ) 
génère une nouvelle date, corrigée (en plus ou en moins) du nombre de mois spécifié…  
Cette fonction est précieuse pour calculer une date d’échéance !    Exemples :   
MOIS.DECALER ( DATEVAL("15/01/93" ; 1 )    égale    34015    ou    15/02/93 
MOIS.DECALER ( DATEVAL("31/03/93" ; - 1 )    égale     34028    ou    28/02/93 

NB.JOURS.OUVRES ( date_départ ; date_fin ; jours_fériés ) 
Renvoie le nombre de jours ouvrés entiers compris entre  date_départ  et  date_fin.  
Les jours ouvrés excluent les fins de semaine et toutes les dates identifiées comme étant des jours fériés. 
Utilisez NB.JOURS.OUVRES pour calculer les charges salariales au prorata du nombre de jours travaillés. 
 

jours_fériés   représente une plage facultative d'une ou plusieurs dates à exclure du calendrier des jours 
de travail, comme les jours fériés ou d'autres jours contractuellement chômés. 
Exemple : NB.JOURS.OUVRES ( DATEVAL("01/10/93") ; DATEVAL("01/12/93") ; DATEVAL("11/11/93"))   égale   43 

FIN.MOIS(date_départ;mois) 
Renvoie le numéro de série de la date du dernier jour du mois précédant ou suivant date_départ du nombre 
de mois indiqué. Utilisez FIN.MOIS pour calculer des dates d'échéance tombant le dernier jour du mois. 
mois   représente le nombre de mois avant ou après date_départ. Une valeur de mois positive donne une 
date future, tandis qu'une valeur négative donne une date passée. 

Exemples : FIN.MOIS(DATEVAL("01/01/93"); 1) égale 34028 ou 28/2/93 
FIN.MOIS(DATEVAL("01/01/93"); -1) égale 33969 ou 31/12/92 

Objectif Fonction, syntaxe et exemples 

Pour prévoir une formule conditionnelle, il faut utiliser la fonction SI… 

Première syntaxe et exemple  (on peut ne pas prévoir de cas « non ») :  

SI (condition ; expression )          exemple :              =SI ( B23 < 0 ; "Attention !" )  
pour faire de la cellule un « drapeau »  (flag)  qui signale fortement un problème important,  sans rien faire sinon… 

Deuxième syntaxe  (souvent, on prévoit un cas « oui » et un cas « non ») :  

SI (condition ; expression si la condition est  vraie ; expression si cond. fausse ) 
Exemple :      =SI ( CATotal = 0 ; 0 ; B23 / CATotal ) 

Ici, pour éviter la division par zéro  qui génère un message d’erreur,  on prévoit deux cas :  
une constante (zéro) en guise de calcul,  ou une formule (calcul de pourcentage). 

Troisième syntaxe  (quelquefois, on est obligé  d’imbriquer les SI  les uns dans les autres…) : 

Exemple : une entreprise consent une ristourne de fin d’année à ses meilleurs clients en fonction du CA cumulé de 
l’année : si le CA est inférieur à 10 000 F, pas de ristourne.  Si le CA est compris entre 10 000   (inclus) et 
20 000 (non compris), ristourne de 5% du CA. Si le CA égale ou dépasse 20 000  sans atteindre 30 000 F, 
ristourne de 8%. Si le CA est égal ou supérieur à 30 000 sans égaler 40 000 F, ristourne de 10%, ristourne 
de 15% dans les autres cas. 

 

=SI(CA<10000;0; SI(CA<20000;0,05; SI(CA<30000;0,08; SI(CA<40000;0,1;0,15) ))) 
 

Ici, on imbrique les SI :  un SI complet est placé dans le cas « non »  d’un autre SI…. 
Pour simplifier la réflexion préalable,  il est fortement recommandé de dessiner un petit graphe  (segment de droite),  

voire même un organigramme  (c’est vite fait et ça montre bien les enchaînements) : 

 

 

 

 

 

 

Un autre schéma, classique et assez clair, est l’organigramme : 

 
Notez que le cas non est plus difficile à lire que le cas oui : chaque comparateur a un 
inverse, et pas exactement celui qu’on imagine au premier abord : le cas non de supé-
rieur  n’est pas inférieur … mais plutôt inférieur ou égal !…  Attention donc à bien 
traiter l’égalité au seuil  comme il faut : lisez bien l’énoncé  et  soyez rigoureux. 
 

 comparateur  inverse  
 = <>  
 <> =  
 > <=  
 >= <  
 < >=  
 <= >  

 

CA < 
10 000 

 ? 

oui  non  

CA < 
20 000 

 ? 

oui  non  

CA < 
30 000 

 ? 

oui  non  

CA < 
40 000 

 ? 

oui  non  

Taux  
= 0 

Taux  
= 0,05 

Taux  
= 0,08 

Taux  
= 0,1 

Taux  
= 0,15 

CA >= 
40 000 

 ? 
oui  non  

CA >= 
30 000 

 ? 

oui  non  

CA >= 
20 000 

 ? 
oui  non  

CA >= 
10 000 

 ? 

oui  non  

Taux  
= 0,15 

Taux  
= 0,1 

Taux  
= 0,08 

Taux  
= 0,05 

Taux  
= 0 

On peut partir du seuil le plus petit  
ou bien du seuil le plus grand,  
comme bon nous semble…  

à condition de  

bien choisir le comparateur… 

∞∞∞∞ 

15% 10% 8% 0% 5% 

10 0000 0 20 0000 30 0000 40 0000 

Toujours dessiner un petit croquis  (à la main ! vite fait…)  pour illustrer les seuils et les actions : 

Tirer un trait  qui va de 0 à l’infini  (∞),    noter les seuils  (inclus) d’un côté du trait    (seuils à égaler ou dépasser),   
  et noter les actions  de l’autre côté du trait  : 

Objectif Fonction, syntaxe et exemples 

EXTRAIRE UNE VALEUR D’UNE TABLE 
 

Une des fonctions les plus importantes dans un tableur est la fonction RECHERCHEV,  qui permet de 
récupérer dans une table une information stable,  à partir d’une information saisie. 

Excel propose 2 syntaxes pour 2 besoins très différents : 
 

RECHERCHEV ( valeur connue ; Table ; ColÀRendre ; FAUX ) 

pour les tables-fichiers (clients, fournisseurs, produits, agences, voyages, etc.) 
Le 4ème argument (FAUX) refuse les approximations : aucune différence ne doit être acceptée entre la valeur connue et 

ce qui est trouvé dans la 1ère colonne de la table. 

 

RECHERCHEV ( valeur connue ; Table ; ColÀRendre ; VRAI ) 

pour les tables-seuils et taux (taux de remises, de commissions, de primes, de TVA, etc.)  
Le 4ème argument (VRAI) accepte les approximations : il est tout à fait possible (et même probable) que la valeur 

connue ne corresponde pas exactement à un des seuils de la 1ère colonne de la table. 
 

J’ai mis des espaces  autour des points-virgules pour la lisibilité , mais en fait, il ne faut JAMAIS  taper d’espace dans les formules  ! 

Voyons d’abord les tables-fichiers : je veux saisir un code  
et récupérer les infos correspondant exactement à ce code dans une table 

Soit une table de référence, 
mémorisée dans une feuille à 

part, et nommée Table-
Clients : 

La table (colonnes entières ! et 

non pas qqs lignes) est nommée. 

Un cas zéro est prévu, pour 
éviter des tests inutiles dans 

la formule de recherche.  

Soit une autre feuille (ci-contre), 
dans laquelle je dois seulement 
saisir le numéro du client (15 par 
exemple) :  

Excel est capable de récupérer 
automatiquement toutes les in-
fos relatives au client n°15 ! 

Si je me trompe (en tapant 10 ou 
13 qui sont des codes inexistants 
dans ma table Clients), Excel me 
signale que les valeurs sont 
Non Accessibles par un messa-
ge #NA. 

Tant que rien n’est saisi en F4, 
Excel cherche zéro (contenu de 
F4), le trouve (en début de table) 
et affiche les données corres-
pondantes (des zéros ou des es-
paces pour garder un affichage 
propre). Mon "cas zéro" dans ma ta-
ble me permet d’économiser de nom-
breux SI pour garder des lignes vides 
tant que rien n’est saisi !  

 
 

 
 

En tapant 15 en F4 (F4 est nommée CodeClientSaisi),  
D6 affiche toute seule « Ets Durand »… C’est magique !!! 

Une autre formule du même tonneau  
permet d’afficher également  « 1 cours Lieutaud » en dessous… 

Objectif Fonction, syntaxe et exemples 

Voyons maintenant  les tables-seuils et taux : je veux saisir un nombre 
et récupérer dans une table la valeur correspondant à  la tranche  à laquelle appartient le nombre. 

Soit une table de référence, nommée TableTauxRistournes qui contient les règles d’attribution des ristour-
nes à mes meilleurs clients : la formule exploitera cette table  

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 
Inutile ici de nommer des colonnes entières, car les perspectives de croissance de la table sont faibles… 

Tant que rien n’est saisi en B6, Excel cherche zéro 
(contenu de B6), le trouve (en début de table, D7) et 
affiche la valeur correspondante (un zéro, ce qui 
n’affiche rien si j’ai désactivé l’affichage des zéros 
avec Outils Options…). 

En tapant 32 400 en B6  (B6  est nommée CA),  
B7 affiche toute seule « 0,1 »… C’est simple !!! 

L’outil « Pourcentage » multipliera ce résultat par 
100 et lui ajoutera le signe %… Avec l’outil « Ajout 

de décimale », on termine la mise en forme. 

Avantages formidables de la fonction RECHERCHEV par rapport à une imbrication de fonctions SI : 

• Grande simplicité de mise en œuvre : la fonction RECHERCHEV, une fois apprivoisée, est beaucoup plus 
simple à taper et à débuguer, plus lisible, qu’un enchevêtrement de SI… 

• De cette simplicité découlent rapidité et fiabilité !!!    Il n’y a pas photo !… 
• On ne peut pas imbriquer les SI sans limites !   Alors que RECHERCHEV est sans limites ! 
• Par ailleurs, la formule qui utilise RECHERCHEV peut être protégée sans gêner l’utilisateur novice (puisqu’on 
ne la modifiera jamais : seule la table fait l’objet des MAJ éventuelles), alors qu’une formule avec des SI 
imbriqués pose le problème de la déprotection nécessaire, en plus de la complexité de la correction, ce 

qui rend, en fait, les MAJ inaccessibles à un novice !  

• Ensuite, la mise à jour éventuelle des règles (nouveaux seuils, nouveaux taux…) est très simple (donc rapi-
de et fiable) directement, clairement, et une seule fois !, dans la table (sans toucher aux formules), alors 
qu’elle est complexe (donc lente et risquée, nécessitant déprotection, réflexion, correction dans le fatras 
de la formule, recopie sans oubli sur les autres formules, reprotection, tests…) avec des fonctions SI !      Il 
faut être complètement masochiste pour imbriquer des SI dans les cas simples.  

Seul cas dans lequel il faut choisir d’imbriquer des SI : taux appliqués cumulativement : quand on applique un 
1er taux à la 1ère tranche PLUS un 2ème taux à la 2ème tranche, PLUS un 3ème taux à la 3ème tranche, etc.. 

Dans ce cas, il faudra se résoudre à imbriquer les SI… ou bien construire une astucieuse imbrication de tables 
et de RECHERCHEV, mais on perdra une partie des qualités de simplicité liées à la fonction (les qualités de 

simplicité de MAJ restant intactes). 

Objectif Fonction, syntaxe et exemples 

Les applications de l’outil RECHERCHEV sont infinies : tous les tableaux du monde (…) offrent un ter-
rain d’application de cette indispensable fonction (j’exagère à peine…). 

Tous les utilisateurs d’Excel devraient connaître cette fonction sur le bout des doigts… Il faut être ca-
pable de monter une fonction Recherche à tout moment, sans mode opératoire, aussi simplement qu’on fait 
intellectuellement une addition, sans calculette… 

Ceci est ESSENTIEL.    Si on ne maîtrise pas RECHERCHEV, on est débutant sous Excel. 
Syntaxe reformulée : 
RECHERCHEV (quoi chercher ; où chercher ; quoi rendre ; accepter ou pas les approximations) 

ou      RECHERCHEV (valeur unique ; table de recherche ; n° colonne ; faux ou vrai) 

En français, on demande à Excel :   
« Cherche cette valeur que je connais  

dans la première colonne de la table désignée,  
et rends-moi ce que tu trouves dans la colonne x de la table,  

sans accepter de différence entre ce que je cherche et ce que tu trouves dans la 1ère colonne de la table ». 
On a vu que la fonction RECHERCHEV, utilisée dans une table-fichier, peut ne pas trouver dans la premiè-
re colonne de la table ce qui a été saisi par l’opérateur (ceci va arriver lors de chaque erreur de saisie, par exemple).   

RECHERCHEV, dans ce cas, affiche un message #NA  qui signale le problème…  #NA pour dire  
« Non Accessible »…   Mais ce message est bien obscur pour un opérateur peu expérimenté. 

Si vous voulez aider un opérateur novice, si ce n’est pas vous qui allez utiliser le tableau,  vous avez in-
térêt à intercepter cette erreur et à ordonner l’af fichage d’un message plus clair :  vous allez utiliser la 
fonction ESTNA  (prononcez É-AINA…),  qui s’utilise uniquement en association avec la fonction SI : 

Syntaxe :     ESTNA ( expression )      rend       VRAI  ou   FAUX,  
 (ce qui fait que cette fonction n’a d’intérêt qu’à l’intérieur d’une condition de fonction SI) 

Exemple : 

=SI ( ESTNA ( RECHERCHEV ( CodeClientSaisi ; TableClients ; 2 ; FAUX ) ) ; "Client inexistant" ;  
RECHERCHEV ( CodeClientSaisi ; TableClients ; 2 ; FAUX ) ) 

En français, on dit à Excel :  
si mon bloc-recherche rend #NA, alors affiche un message clair, sinon affiche le résultat du bloc-recherche. 

Dans cette formule assez longue, vous remarquerez que le bloc-Recherche est tapé 2 fois … Pas moyen de faire au-
trement, mais la saisie de la formule est quand même très simple si on connaît bien son presse-papier et ses raccour-
cis claviers : je tape le début de la formule (toute la première ligne de l’exemple ci-dessus), et quand c’est le moment de 
taper mon deuxième bloc-Recherche, je sélectionne le premier avec la souris, je tape CTRL + C pour le mettre au pres-
se-papier, je me replace en fin de formule avec la touche Fin , puis je tape CTRL + V pour vider le presse-papier… C’est 
beaucoup plus long à écrire qu’à faire… Essayez, vous serez vite convaincu de l’efficacité de cette manipulation… 

La cerise sur le gâteau de l’ergonomie (c’est-à-dire du confort) de mon tableau, c’est d’associer un format conditionnel 
à mon message d’erreur : si, en plus d’afficher un message clair et lisible, ma formule passe en blanc gras sur fond 
rouge dès qu’elle affiche 'Client inexistant", c’est spectaculaire et très soigné…  On va vous féliciter, c’est sûr !…  

Toujours en complément de la fonction RechercheV,  il faut connaître INDEX et EQUIV qui prennent en charge les cas 
tordus dans lesquels Recherchev ne fonctionne pas… En gros, EQUIV retrouve une valeur connue dans un vecteur (li-
gne ou colonne) et rend un rang  (un numéro dans la série de cellules alignées)… alors que INDEX est capable 
d’utiliser ce rang pour trouver la cellule équivale nte (de même rang) dans un autre vecteur  (ligne ou colonne or-
ganisée de la même façon que le 1er vecteur).   Cette combinaison permet de chercher dans une feuille  et  de  
trouver dans  une autre  feuille ,  même dans un autre classeur  si on veut !   On n’a plus la contrainte de la table 
unique, avec sa 1 ère colonne  qui est la seule colonne de recherche possible ! 
Syntaxe : EQUIV ( valeur_cherchée ; zone_de_recherche ; type )  

permet de localiser (par un rang) une valeur dans une ligne ou une colonne.   
Si type=0, EQUIV s’arrête sur la première occurrence trouvée : le vecteur peut être en désordre.   
Si type = 1, il faut que le vecteur soit en ordre croissant.   (Type = -1, vecteur trié décroissant). 
INDEX ( zone_d’extraction ; indice_Ligne ; indice_Colonne )  
permet de piocher dans un tableau grâce à 2 indices numériques (un seul si la zone n’a qu’une dimension) 
INDEX ( vecteur_valeur_à_rendre ; EQUIV ( valeur_connue ; vecteur_de_recherche ) ) 

Exemple : INDEX ( SalairesNets ; EQUIV ( NomSalariéSaisi ; NomsSalariés ; 0 ) ) 
On cherche et on trouve dans des colonnes placées où l’on veut : pas de contrainte « Table et 1ère colonne »)… 

Objectif Fonction, syntaxe et exemples 

Certains calculs génèrent  

des  micro-décimales : 
Multiplier un prix à deux décimales par un taux 
de TVA (à 3 décimales) donne un résultat à 5 
décimales.  

Seules les 2  1ères  décimales sont significatives.  

Les 3 autres sont des microdécimales à éliminer.  

Mais il faut les éliminer en profondeur avec une 
fonction ARRONDI, et pas seulement en appa-
rence avec une commande Format ! 

Sinon, vous allez constater une différence entre 
la somme affichée par Excel (qui compte les mi-
crodécimales) et la somme effectuée avec une 
calculette (qui ne les compte pas)  !!! et vos partenai-
res ne vont pas aimer cette différence de cen-
times qui fait naître la suspicion !!!  

 
ARRONDI ( expression ; nbdécimales ) 

expression désigne une formule ou une adresse  
ou autre moyen de générer une valeur 

• Si nbdécimales est supérieur à 0 (zéro), expression est arrondie 
au nombre de décimales indiqué (au plus proche). 

• Si nbdécimales est égal à 0, expression est arrondie au nombre 
entier le plus proche. 

• Si nbdécimales est inférieur à 0, expression est arrondie à gau-
che de la virgule (à la dizaine la + proche,  la centaine, etc.) 

 
Exemples 

ARRONDI(2,15; 1) égale 2,2 
ARRONDI(2,149; 1) égale 2,1 
ARRONDI(-1,475; 2) égale -1,48 
ARRONDI(24,45; -1) égale 20 

On peut forcer l’arrondi au supérieur ou à 
l’inférieur :   

Les règles sont les mêmes qu’avec ARRONDI : on peut 
arrondir à l’entier en spécifiant zéro en 2ème argument.  

On peut aussi arrondir à la dizaine, à la centaine, etc. 

 

ARRONDI.SUP (expression; nbdécimales ) 

ARRONDI.INF (expression; nbdécimales ) 

Pour extraire la partie entière d’un nom-
bre,  on peut utiliser la fonction ENT  
  
 ou la fonction TRONQUE… 

 
Ces deux fonctions se ressemblent, mais ont 
des réactions différentes  en dessous de zéro. 

ENT ( expression ) Arrondit un nombre  
à l'entier immédiatement inférieur. 

Exemples : ENT(8,9) égale 8 
ENT(-8,9) égale –9 

TRONQUE ( expression ; no_chiffres ) Tronque un 
nombre en supprimant la partie décimale de ce nombre de 
sorte que la valeur renvoyée par défaut soit un nombre entier. 

Exemples : TRONQUE(8,9) égale 8 
TRONQUE(-8,9) égale -8 

Pour calculer le reste d’une division,  
il faut utiliser la fonction MOD 

MOD ( expression ; diviseur )  
Renvoie le reste de la division de l'argument expression 
par l'argument diviseur.  
(Le résultat est du même signe que diviseur.) 

Exemple : MOD(3 ; 2)   égale   1 
Pour faire la somme seulement  

de certaines cellules d’une plage,  
il faut absolument connaître  

la très précieuse fonction SOMME.SI ! 
Le troisième argument est facultatif : on ne le 
remplit pas quand la zone à tester est la 

même que la zone à sommer.  

La fonction NB.SI est également souvent 
très utile !  Elle permet de dénombrer les 
cellules d’une colonne qui répondent à 

une condition (simple).  

Cette fonction n’a que 2 arguments car la 
zone à tester est toujours la même que la zo-

ne à dénombrer. 

Syntaxe :      Attention au test : guillemets, sauf si simple égalité à une zone nommée ! 
SOMME.SI ( zone à tester ; test ; zone à sommer ) 
Exemple : SOMME.SI ( Catégorie ; "cadres " ; Salaires) 

va dégager la masse salariale des cadres de la colonne de 
tous les salaires, pourvu qu’une colonne voisine contienne de 
quoi faire la distinction (ici, c’est une colonne qui affiche le 
texte de la catégorie en clair, mais ce pourrait être n’importe 

quel autre indice codé, numérique ou autre…) 

Exemple :  SOMME.SI ( MontantVente ; ">=10000" )  
va dégager de la colonne des montants individuels de cha-

que vente, le CA global des « grosses » ventes  (>10 000F)… 

Exemple :  SOMME.SI ( TxTVATVA ; TxTVA1 ; Montant ) 
va sommer les seuls montants pour lesquels le taux de TVA est 

égal au taux contenu dans la cellule nommée TxTVA1. 

Syntaxe : NB.SI ( zone à tester et dénombrer ; test ) 
Exemples : =NB.SI ( notes ; ">=10" ) compte les bonnes 

notes de la dernière interrogation écrite… 

Autre exemple : =NB.SI ( flag ; 1 ) compte le nombre de cellules de la colonne "flag" (drapeau) qui signalent un pro-
blème (en affichant le chiffre 1)…  Cette astuce permet de dénombrer des lignes avec des conditions complexes 
(plusieurs SI imbriqués qui débouchent sur la valeur 1 si le problème est constaté), ce que ne permet pas  NB.SI  seule. 

Attention  ! dans la réalité, 
bien sûr, on ne tapera pas 
des chiffres « en dur »  
comme dans les exemples : 
on tapera une adresse  ou 
une formule  de calcul… 



Formules Excel : aide-mémoire des opérateurs, des pointeurs et des fonctions 15/18 

Objectif Fonction, syntaxe et exemples 

 
Pour manier les chaînes de caractères,   
il faut connaître les  fonctions texte. 

 

On a souvent besoin de les utiliser en-
semble, par des combinaisons astucieu-
ses… 

 

On n’oubliera pas que :  

l’opérateur de concaténation est le  &  
(on dit « éperluette » ou « et commercial ») 

DROITE ( texte ; nbcar )   :  Renvoie les x derniers carac-
tères à droite d'une chaîne de texte.  

Exemple :  DROITE ( "Prix de vente"; 5 ) égale "vente" 

GAUCHE ( texte ; nbcar ) : idem à l’envers… 

NBCAR ( texte )   :  Longueur (nb de car) d'une chaîne. 
Exemple :   NBCAR ( "Prix de vente" ) égale 13 

STXT ( texte ; numdépart ; nbcar ) 
Renvoie une sous-chaîne de n caractères extraits d'un 
texte à partir de la position que vous avez spécifiée. 

Exemple : STXT ("Cours moyen"; 1; 5 ) égale "Cours" 
STXT ("Cours moyen"; 7; 20 ) égale "moyen" 

Dans les exemples ci-dessus,  on connaît le point de départ et la longueur de la chaîne voulue, 
or ces infos sont souvent variables…  La fonction suivante est donc essentielle ! Elle sert à loca-
liser un caractère donné dans la chaîne : 

CHERCHE ( texte_cherché ; texte ; no_départ ) 
Renvoie le numéro du caractère au niveau duquel un texte cherché est initialement reconnu lors d'une 
lecture de gauche à droite.  Vous pouvez utiliser les caractères génériques (jokers), le point d'interroga-
tion (?) et l’étoile (*) dans l'argument texte_cherché. Le point d'interrogation correspond à un caractère 
unique quelconque et l’étoile correspond à une séquence de caractères quelconque. Si vous voulez 
trouver réellement un point d'interrogation ou un astérisque, tapez un tilde (~) devant ce caractère. 

Exemple : Si la cellule B17 contient le mot « profit » et la cellule A14 l'expression « Pertes et profits » : 
   CHERCHE ( $B$17 ; $A$14 )   égale   11 

Utilisez la fonction CHERCHE pour trouver la position d'un caractère ou d'une chaîne de texte dans une autre 
chaîne de texte de façon à pouvoir utiliser ensuite l'une des fonctions STXT (ou SUBSTITUE) pour modifier le texte. 

TROUVE ( texte_cherché ; texte ; no_départ ) 
la fonction TROUVE, à la différence de la fonction CHERCHE, respecte les majuscules et les 

minuscules et n'admet pas de caractère générique. 

MINUSCULE ( texte ),   MAJUSCULE ( texte ),   NOMPROPRE ( texte )  sont faciles à comprendre seul… 

SUBSTITUE ( texte ; ancien_texte ; nouveau_texte ; no_position ) 

Exemples : SUBSTITUE("Données Ventes"; "Ventes"; "Coût") égale "Données Coût" 
SUBSTITUE("Trimestre 1 - 1991"; "1"; "2"; 1) égale "Trimestre 2 - 1991" 
SUBSTITUE("Trimestre 1 - 1991"; "1"; "2"; 3) égale "Trimestre 1 - 1992" 

Exemple de combinaison : Je viens de récupérer chez un ami un classeur dans lequel  la même colonne 
contient à la fois les noms et les prénoms des gens, et dans le mauvais ordre…  Or je veux récupérer ces 
infos (sans les resaisir, bien sûr),  en mettant  le prénom avant le nom   (lui-même en majuscules) : 

 

 

 

 

 

 

 

 
 

Bien sûr, le découpage de la formule en 7 colonnes est là pour expliquer progressivement les possibilités… 
On pourrait aussi bien arriver au même résultat en une seule formule  (un peu moins digeste, il est vrai…) : 
=STXT(A3;1;CHERCHE(" ";A3))&" "&DROITE(A3;NBCAR(A3)-CHERCHE(" ";A3;CHERCHE(" ";A3)+1))&" " 
&MAJUSCULE(STXT(A3;CHERCHE(" ";A3);CHERCHE(" ";A3;CHERCHE(" ";A3)+1)-CHERCHE(" ";A3))) 

Objectif Fonction, syntaxe et exemples 

Excel offre des outils nombreux pour les 

calculs  financiers .   

Nous nous limiterons aux fonctions simples 
qui tournent autour des simulations d’em-
prunts et d’épargne. 

 

Toutes les situations d’emprunt  
se définissent avec 5 variables :  

 

� VA = valeur actuelle  
Ce que représente aujourd’hui l’ensemble des rem-
boursements futurs.  
Par exemple, le montant de l’emprunt pour le prêteur. 

� VC = valeur future 
Valeur capitalisée, avec les intérêts 

� VPM = montant des remboursements 
Valeur des montants périodiques (par an, par 
mois…) 

� NPM = nombre des remboursements 
Nombre des périodes pour des remboursements 
constants, à un taux d’intérêt constant. 
Si on rembourse tous les mois un emprunt 
sur 4 ans, NPM = 48  (4 ans x 12 mois par an). 

� TAUX = taux de l’emprunt 
Taux d’intérêt par période (année, mois…) 
Le taux doit être en rapport avec les périodes : 
Si on rembourse tous les mois, un taux annuel 
devra être divisé par 12 pour donner à Excel un 
taux mensuel. 

 

Il faut connaître 4 variables pour calculer 
la cinquième,  l’inconnue…  

 

Excel propose donc 5 fonctions :  
une fonction par cas de figure,   

une fonction par variable inconnue . 

 

Remarques générales (pour toutes les fonctions) : 

• Les sommes reçues  sont positives , alors 
que les sommes versées  sont négatives .  

On peut se placer du point de vue de 
l’emprunteur ou au contraire du prêteur… il 
suffit d’adapter les signes… 

 

• L’argument Type = 0 sert à dire que les 
remboursements ont lieu en fin de pé-
riode.      Type = 1 => début de période. 

 

� Combien faudra-t-il rembourser à chaque période ? 

VPM ( taux ; npm ; va ; vc ; type ) 

Exemple  : cette formule renvoie la valeur du rem-
boursement mensuel d'un emprunt de 10 000 F au 
taux annuel de 8 % remboursable en 10 mois :  
=VPM ( 8%/12 ; 10 ; 10000 )  égale  –1 037,03 F 

� Combien faut-il prévoir de temps pour rembourser ? 

NPM ( taux ; vpm ; va ; vc ; type ) 

Vc représente la valeur à atteindre après le dernier 
versement : 0 si c’est un emprunt , un nombre si 
c’est une épargne. 

Exemple  : la formule suivante renvoie le nombre de 
mensualités pour le remboursement mensuel d'un 
emprunt de 10 000 F au taux annuel de 12 % à raison 
de 1000 F par mois :   
=NPM(12%/12; -1000; 10000; 0; 1) égale 11  
(11 mensualités) 

� Combien vaudra mon épargne ? 

VC ( taux ; npm ; vpm ; va ; type ) 

Renvoie la valeur future d'un investissement à rem-
boursements périodiques et constants, et à un taux 
d'intérêt constant. 

Exemple  : Supposons que vous vouliez économiser de 
l'argent pour financer un projet spécifique qui sera mis 
en oeuvre dans un an. Vous déposez 1 000 F sur un 
compte d'épargne qui vous rapporte 6 % d'intérêts par 
an, capitalisés mensuellement, ce qui représente un in-
térêt mensuel de 6%/12, soit 0,5%. Vous envisagez de 
déposer 100 Francs au début de chaque mois pendant 
les 12 mois à venir. Quel sera le montant de votre 
épargne au bout des 12 mois ? 
=VC( 0,5% ; 12 ; -100 ; -1000 ; 1) égale  2 301,40 F 

� Quel est le taux d’un emprunt ? 

TAUX ( npm ; vpm ; va ; vc ; type ; estimation ) 

La formule suivante donne le taux d'intérêt d'un em-
prunt de 80 000 F sur 4 ans (48 = 4 x 12 mois) avec 
des remboursements mensuels de 2 000 F : 
=TAUX ( 48 ; -2000 ; 80000 )  égale  0,77 % 

� Quelle est la valeur actuelle d’un investissement ? 

VA ( taux ; npm ; vpm ; vc ; type ) 
Exemple : une compagnie financière vous propose un inves-
tissement qui vous rapportera 5 000 F à la fin de chaque mois 
pendant les vingt prochaines années. Le montant de l'inves-
tissement est de 600 000 F et le taux d'intérêt du marché est 
de 8 % par an. En utilisant la fonction VA, vous découvrez que 
la valeur actuelle de la suite d'annuités de 5 000 F est : 

=VA ( 0,08/12 ; 12*20 ; 5000 ; ; 0 )  égale  -597 771,50 F 
Le résultat est négatif car il représente un décaissement, ce que vous 
devriez payer. La valeur actuelle de la suite d'annuités (597 771,50 F) 

est inférieure à la somme que vous devez payer (600 000 F) pour réali-
ser cet investissement. Il n'est donc pas intéressant de souscrire à la 

proposition de la compagnie. 

